Four Problems with Policy-Based Constraints and How to Fix Them

Dillan Mills
Synopsys, Inc.
Maricopa, AZ
dillan@synopsys.com

Chip Haldane
The Chip Abides, LLC
Gilbert, AZ
chip@thechipabides.com

Abstract

This paper presents solutions to problems encountered in the implementation of policy classes for SystemVerilog constraint layering. Policy classes provide portable and reusable constraints that can be mixed and matched into the object being randomized. There have been many papers and presentations on policy classes since the original presentation by John Dickol at DVCon 2015. The paper addresses three problems shared by all public policy class implementations and presents a solution to a fourth problem. The proposed solutions introduce policy class inheritance, tightly pair policy definitions with the class they constrain, reduce the expense of defining common policies using macros, and demonstrate how to treat policies as disposable and lightweight objects. The paper concludes that the proposed solution improves the usability and efficiency of policy classes for SystemVerilog constraint layering.

I. Constraints and Policy Class Review

Random objects and constraints are the foundational building blocks of constrained random verification in SystemVerilog. The simplest implementations embed fixed constraints within a class definition. Embedded constraints lack flexibility; all randomized object instances must meet these requirements exactly as they are written.

In-line constraints using the **`with`** construct offer marginally better flexibility. Although these external constraints allow greater variability of random objects, their definitions are still fixed within the calling context. Furthermore, all in-line constraints must be speficied within a single call to **`randomize()`** .

Policy classes are a technique for applying SystemVerilog constraints in a portable, reuasable, and incremental manner, originally described by John Dickol [**?**] [**?**]. The operating mechanism leverages an aspect of "global constraints," the simultaneous solving of constraints across a set of random objects. Randomizing an object that contains policies also randomizes the policies. Meanwhile, the policies contain a reference back to the container. Consequently, the policy container is constrained by the policies it contains. Dickol's approach is illustrated by the following code.

```
class policy_base#(type ITEM=uvm_object);
    ITEM item;

    virtual function void set_item(ITEM item);
        this.item = item;
    endfunction
endclass

class policy_list#(type ITEM=uvm_object) extends policy_base#(ITEM);
    rand policy_base#(ITEM) policy[$];

    function void add(policy_base#(ITEM) pcy);
        policy.push_back(pcy);
    endfunction

    function void set_item(ITEM item);
        foreach(policy[i]) policy[i].set_item(item);
    endfunction
endclass
```

Fig. 1: The **`policy_base`** and **`policy_list`** classes

The first two classes provide the core definitions for policies. The **policy_base** implements the hook back to the policy container, and the **policy_list** enables related policies to be organized into groups. Both of these classes are parameterized by a container object type, so a unique specialization will be required for each policy container class that needs to be constrained.

Next, **addr_txn** is an example policy container—a generic transaction with a random address and size that implements policies.

```
class addr_txn;
    rand bit [31:0] addr;
    rand int        size;
    rand policy_base#(addr_txn) policy[$];

    constraint c_size {size inside {1, 2, 4};}

    function void pre_randomize;
        foreach(policy[i]) policy[i].set_item(this);
    endfunction
endclass
```

Fig. 2: The **addr_txn** class

The next three classes implement some policies for the **addr_txn** class. A list of valid address ranges can be stored in the **ranges** array. The **addr_permit_policy** will choose one of the ranges at random and constrain the address to be within the range, while the **addr_prohibit_policy** will exclude the address from randomizing inside any of the ranges in its list.

```
class addr_policy_base extends policy_base#(addr_txn);
    addr_range ranges[$];

    function void add(addr_t min, addr_t max);
        addr_range rng = new(min, max);
        ranges.push_back(rng);
    endfunction
endclass

class addr_permit_policy extends addr_policy_base;
    rand int selection;

    constraint c_addr_permit {
      selection inside {[0:ranges.size()-1]};

        foreach(ranges[i]) {
            if(selection == i) {
                item.addr inside {[ranges[i].min:ranges[i].max - item.size]};
            }
        }
    }
endclass

class addr_prohibit_policy extends addr_policy_base;
    constraint c_addr_prohibit {
        foreach(ranges[i]) {
            !(item.addr inside {[ranges[i].min:ranges[i].max - item.size + 1]});
        }
```

```
    }
endclass
```

Fig. 3: Policies for constraining the **addr_txn** class

The final class shows how policies might be used. The **addr_constrained_txn** class extends **addr_txn** and defines two policies, one that permits the address to be within one of two ranges, and one that prohibits the address from being within a third range. The **addr_constrained_txn** class then passes the local **pcy** list to the parent **policy** queue.

```
class addr_constrained_txn extends addr_txn;
    function new;
        addr_permit_policy     permit = new;
        addr_prohibit_policy   prohibit = new;
        policy_list#(addr_txn) pcy = new;

        permit.add('h00000000, 'h0000FFFF);
        permit.add('h10000000, 'h1FFFFFFF);
        pcy.add(permit);

        prohibit.add('h13000000, 'h130FFFFF);
        pcy.add(prohibit);

        this.policy = {pcy};
    endfunction
endclass
```

Fig. 4: The **addr_constrained_txn** subclass

At this point, an instance of **addr_constrained_txn** can be created and randomized like normal, and the address will be constrained based on the embedded policies.

```
addr_constrained_txn txn = new();
txn.randomize();
```

Fig. 5: Randomizing an instance of **addr_constrained_txn**

Further work on policy-based constraints has been presented since the original DVCon presentation in 2015. Kevin Vasconcellos and Jeff McNeal applied the concept to test configuration and added many nice utilities to the base policy class [**?**]. Chuck McClish extended the concept to manage real number values for User Defined Nettypes (UDN) and Unified Power Format (UPF) pins in an analog model [**?**]. Additionally, McClish defined a policy builder class that was used to generically build multiple types of policies while reducing repeated code that was shared between each policy class in the original implementation.

Although there has been extensive research on policies, this paper aims to address and provide solutions for three issues that have not been adequately resolved in previous implementations. Furthermore, a fourth problem that arose during testing of the upgraded policy package implementation will also be discussed and resolved.

II. Problem #1: Parameterized Policies

The first problem with the above policy implementation is that because policies are specialized using parameterization to the class they constrain, different specializations of parameterized classes cannot be grouped and indexed. This is most apparent when creating a class hierarchy. If you wanted to extend a class and add a new field to constrain then you need a new policy type. The new policy type requires an extra policy list to modify any new attributes in the child class and requires the user to know which layer of the class hierarchy defines each attribute they want to constrain. Imagine a complex class hierarchy with many layers of inheritance and extension, and new constrainable attributes on each layer (one common example is a multi-layered sequence API library, such as the example presented by Jeff Vance [**?**]). Using policies becomes cumbersome

in this case because you need to manage a separate policy list for each level of the hierarchy and know what each list can and cannot contain. For example, extending the **addr_txn** class to create a version with parity checking results in a class hierarchy that looks like this:

```
class addr_txn;
  // ... unchanged from previous example
endclass

class addr_p_txn extends addr_txn;
  rand bit parity;
  rand bit parity_err;
  rand policy_base#(addr_p_txn) addr_p_policy[$];

  constraint c_parity_err {
    soft (parity_err == 0);
    (parity_err) ^ ($countones({addr, parity}) == 1);
  }

  function void pre_randomize;
    super.pre_randomize();
    foreach(addr_p_policy[i]) addr_p_policy[i].set_item(this);
  endfunction
endclass

class addr_constrained_txn extends addr_p_txn;
  function new;
    addr_permit_policy        permit_p = new;
    addr_prohibit_policy      prohibit_p = new;

    // definition to follow in a later example - sets parity_err to the value
↪   provided
    addr_parity_err_policy    parity_err_p = new;

    policy_list#(addr_txn)    addr_pcy_lst = new;
    policy_list#(addr_p_txn)  addr_p_pcy_lst = new;

    permit_p.add('h00000000, 'h0000FFFF);
    permit_p.add('h10000000, 'h1FFFFFFF);
    addr_pcy_lst.add(permit_p);

    prohibit_p.add('h13000000, 'h130FFFFF);
    addr_pcy_lst.add(prohibit_p);

    parity_err_p.set(1'b1);
    addr_p_pcy_lst.add(parity_err_p);

    this.addr_policy   = {addr_pcy_lst};
    this.addr_p_policy = {addr_p_pcy_lst};
  endfunction
endclass
```

Fig. 6: Modified address transaction classes with a parity checking subclass and policies included

This class ends up with a lot of boilerplate code and is not very intuitive to use. The user needs to keep track of what the difference is between each policy list and apply constraints properly. Each additional subclass in a hierarchy will cause

the constrained class to grow further. The solution to this problem is to replace the parameterized policy base with a non-parameterized base and a parameterized extension. We chose to use an interface class as our non-parameterized container for the flexibility it offers over inheriting from a virtual base class—namely that our parameterized base is free to implement other interfaces, and should the need ever arise, other classes might implement the policy interface.

```
interface class policy;
    pure virtual function void set_item(uvm_object item);
endclass

virtual class policy_imp#(type ITEM=uvm_object) implements policy;
    protected rand ITEM m_item;

    virtual function void set_item(uvm_object item);
        if (!$cast(m_item, item)) begin
            `uvm_warning("policy::set_item()", "Item/policy type mismatch")
            this.m_item = null;
            this.m_item.rand_mode(0);
        end
    endfunction: set_item
endclass: policy_imp

typedef policy policy_queue[$];
```

Fig. 7: The **policy** interface class and **policy_imp** class

Implementing a non-parameterized base (in this case, an interface class) enables all policies targeting a particular class hierarchy to be stored within a single common **policy_queue** implemented by the base class of the target class hierarchy. The **uvm_object** class is a good base class when using policies within UVM, as it will likely cover every class you will be randomizing. If working with another framework, a reasonable base class will need to be chosen. Because the **set_item** function no longer receives a parameterized type input, the input needs to be cast to the implementation type. If this cast fails, some form of cleanup is necessary. In the example above, the item is set to **null** and randomization is disabled when the cast fails.

```
class addr_policy extends policy_imp#(addr_txn);
   // ... unchanged from previous example, with updated class extension
endclass

class addr_parity_err_policy extends policy_imp#(addr_p_txn);
   protected bit parity_err;

   constraint c_fixed_value {m_item != null -> m_item.parity_err == parity_err;}

   function new(int parity_err);
      this.parity_err = parity_err;
   endfunction
endclass

class addr_permit_policy extends addr_policy;
   // same as before
endclass

class addr_prohibit_policy extends addr_policy;
   // same as before
```

```
endclass
```

Fig. 8: Address policies updated to use the new policy implementation

The address policies are updated to inherit from **policy_imp** parameterized to the relevant transaction class. Additionally, the constraints are updated to verify the internal item handle is not null.

```
class addr_txn extends uvm_object;
    rand policy_queue policy;
    // policy is replaced with the above. All other members, constraints,
    // and pre_randomize are unchanged from the previous example
endclass

class addr_p_txn extends addr_txn;
    rand bit parity;
    rand bit parity_err;
    constraint c_parity_err {/*...*/}
    // The local addr_p_policy and pre_randomize are removed. Everything
    // else is unchanged from the previous example
endclass

class addr_constrained_txn extends addr_p_txn;
    function new;
        // only a single policy queue is necessary now
        policy_queue pcy;

        addr_permit_policy     permit_p   = new();
        addr_prohibit_policy   prohibit_p = new();
        addr_parity_err_policy parity_err_p;

        permit_p.add('h00000000, 'h0000FFFF);
        permit_p.add('h10000000, 'h1FFFFFFF);
        pcy.push_back(permit_p);

        prohibit_p.add('h13000000, 'h130FFFFF);
        pcy.push_back(prohibit_p);

        parity_err_p = new(1'b1);
        pcy.push_back(parity_err_p);

        this.policy = {pcy};
    endfunction
endclass
```

Fig. 9: Address transaction classes updated to use the new policy implementation

The new address transaction classes are simplified considerably. Only a single policy queue is required in the class hierarchy, and the **addr_constrained_txn** class does not need to keep track of any specialized lists. The **addr_txn** class now subclasses **uvm_object** to provide compatibility with the policy interface.

III. Problem #2: Definition Location

The second problem with policies is "where do I define my policy classes?" This is not a complicated problem to solve, and most users likely place their policy classes in a file close to the class they are constraining. However, by explicitly pairing policy definitions with the class they constrain, users never need to hunt for the definitions or guess where they are located, and the definitions are given an appropriate and consistent home in the environment. This is done by embedding a **POLICIES**

class inside the class being constrained. This class acts as a container for all relevant policies and allows policies to be created by simply referencing the embedded class through the `::` operator.

```
class addr_txn extends uvm_object;
    // class members, constraints, and pre_randomize unchanged from previous
    ↪ example

    class POLICIES;
        class addr_policy extends policy_imp#(addr_txn);
            // ... unchanged from previous standalone class example
        endclass

        class addr_permit_policy extends addr_policy;
            // ... unchanged from previous standalone class example
        endclass

        class addr_prohibit_p_policy extends addr_policy;
            // ... unchanged from previous standalone class example
        endclass
    endclass: POLICIES
endclass

class addr_p_txn extends addr_txn;
    // class members and constraints unchanged from previous example

    class POLICIES extends addr_txn::POLICIES;
        class addr_parity_err_policy extends policy_imp#(addr_p_txn);
            // ... unchanged from previous example
        endclass

        static function addr_parity_err_policy PARITY_ERR(bit value);
            PARITY_ERR = new(value);
        endfunction
    endclass: POLICIES
endclass

class addr_constrained_txn extends addr_p_txn;
    function new;
        policy_queue pcy;

        addr_constrained_txn::POLICIES::addr_permit_policy   permit_p   = new();
        addr_constrained_txn::POLICIES::addr_prohibit_policy prohibit_p = new();

        // policy constraint value setup unchanged from previous example

        pcy.push_back(addr_constrained_txn::POLICIES::PARITY_ERR(1'b1));

        this.policy = {pcy};
    endfunction
endclass
```

Fig. 10: Address transaction classes with embedded policies

In the above code, there were few changes from the previous iteration. All the policy classes were migrated into an embedded **`POLICIES`** class, and a static constructor was created for the parity error policy to simplify its creation. The

addr_p_txn::POLICIES class was set as a child of the **addr_txn::POLICIES** class, which further simplified the usage of the classes in the constraint block. Finally, the **addr_constrained_txn::POLICIES::** accessor is used to access policies in the constraint block.

A nice side effect of embedding policy class definitions within their target classes is that policies have access to protected members of the target class. This enables meta members of the target class (e.g., "knobs" such as **parity_err**) to be defined as protected, preventing external classes from manipulating them directly. In fact, a more advanced use of policies might define all members of a target class as protected, restricting the setting of fields exclusively through policies and reading through accessor functions, thus encouraging maximum encapsulation/loose coupling, which reduces the cost of maintaining and enhancing code and prevents bugs from cascading into classes that use policy-enabled classes.

IV. Problem #3: Boilerplate Overload

The third problem with using policies is that policies are relatively expensive to define since you need at a minimum: a class definition, a constructor, and a constraint. This will be relatively unavoidable for complex policies, such as those defining a relationship between multiple specific class attributes. For generic policies, such as equality constraints (property equals X), range constraints (property between Y and Z), or set membership (keyword **inside**) constraints, macros can be used to drastically reduces the expense and risk of defining common policies. The macros are responsible for creating the specialized policy class for the required constraint within the target class, as well as a static constructor function that is used to create new policy instances of the class. Additional macros are utilized for setting up the embedded **POLICIES** class within the target class. These macros can be used hand in hand with the non-macro policy classes needed for complex constraints, if necessary.

```
// Fixed-value policy class and constructor macro
`define fixed_policy(POLICY, TYPE, field)                                   \
`m_fixed_policy_class(POLICY, TYPE, field)                                  \
`m_fixed_policy_constructor(POLICY, TYPE, field)

`define m_fixed_policy_class(POLICY, TYPE, field)                           \
    class POLICY``_policy extends base_policy;                              \
        protected TYPE m_fixed_value;                                       \
                                                                            \
        constraint c_fixed_value {                                          \
            m_item != null -> m_item.field == m_fixed_value;                \
        }                                                                   \
                                                                            \
        function new(TYPE value);                                           \
            this.m_fixed_value = value;                                     \
        endfunction                                                         \
    endclass: POLICY``_policy

`define m_fixed_policy_constructor(POLICY, TYPE, field)                     \
    static function POLICY``_policy POLICY(TYPE value);                     \
        POLICY = new(value);                                                \
    endfunction: POLICY
```

Fig. 11: Macros for setting up the embedded **POLICIES** class and a fixed value policy

This example includes a **`fixed_policy** macro, which wraps two additional macros responsible for creating a policy class and a static constructor for the class. This **`fixed_policy** example policy class lets you constrain a property to a fixed value. A more complete macro definition can be found in Appendix **??**. The appendix includes **`start_policies** , **`start_extended_policies** , and **`end_policies** macros that are used to create the embedded **POLICIES** class within the constrained class instead of using hard-coded **class** and **endclass** statements. They set up class inheritance as needed and create a local typedef for the **policy_imp** parameterized type.

```
class addr_txn extends uvm_object;
```

```
    // class members, constraints, and pre_randomize unchanged from previous example

    `start_policies(addr_txn)
        `include "addr_policies.svh"
    `end_policies
endclass

class addr_p_txn extends addr_txn;
    // class members and constraints unchanged from previous example

    `start_extended_policies(addr_p_txn, addr_txn)
        `fixed_policy(PARITY_ERR, bit, parity_err)
    `end_policies
endclass

class addr_constrained_txn extends addr_p_txn;
    // ... unchanged from previous example
endclass
```

Fig. 12: Simplified address transaction classes using the policy macros

The base **addr_txn** class has complex policies with a relationship between the **addr** and **size** fields, so rather than creating a policy macro that will only be used once, they can either be left as-is within the embedded policies class, or moved to a separate file and included with **`include** as was done here to keep the transaction class simple. The child parity transaction class is able to use the **`fixed_policy** macro to constrain the **parity_err** field. The constraint block remains the same as the previous example.

V. Problem #4: Unexpected Policy Reuse Behavior and Optimizing for Lightweight Policies

A fourth problem we noticed in our initial usage of policy classes was occasional unexpected behavior when attempting to reuse a policy. It is hard to quantify and we never spent time to create a reproducible test case, but some of the policies behaved like they "remembered" being randomized when used repeatedly and wouldn't reapply their constraints during subsequent **randomize** calls. Rather than spending time to diagnose and fix this issue, we were focusing on transitioning our policy approach to a "use once and discard" approach, where we would treat a policy as disposable and apply fresh policy instances before re-randomizing a target object. This improvement happened to resolve the randomization issue as well, so we never had a reason to circle back.

Fortunately, creating new policy instances for each randomization is not prohibitively expensive. We anticipated that we would need hundreds or thousands of policies per test, so we aimed to minimize the memory requirements and streamline functionality as much as possible. To minimize the footprint of policies, we chose to implement them as simple classes rather than extensions of **uvm_object** (the policy container itself still inherits from **uvm_object**). We duplicated some useful capabilities such as a basic copy routine, while omitting more expensive functionality such as factory compatibility or object comparison. The buggy randomization behavior implied we needed to pass in fresh policy instances each time we randomized an object. Two features helped to accomplish this: static constructor functions and a **copy** method.

```
static function addr_parity_err_policy PARITY_ERR(bit value);
    PARITY_ERR = new(value);
endfunction
...
pcy.push_back(addr_constrained_txn::POLICIES::PARITY_ERR(1'b1));
```

Fig. 13: Example static constructor function from the **PARITY_ERR** policy class

Using static functions to create new instances of the policies provides an easy way to create and use a policy with a single function call. The policy is created and immediately pushed onto the policy queue, then discarded and recreated the next time it is needed.

```
interface class policy;
    ...
    pure virtual function policy copy();
endclass: policy
```

Fig. 14: The **copy** method in the **policy** interface class

Adding a **copy** method to the policy classes provides a way for a policy to return a fresh copy of itself with the same configuration. This is useful when a policy needs to be reused multiple times. The policy can be copied and pushed onto the policy queue multiple times, or onto other policy queues, and each copy will constrain the field separately. The **copy** method is **pure virtual** , so it must be implemented by each policy class.

After migrating to this pattern of reuse, the issues we were seeing with unapplied constraints disappeared.

VI. More Improvements to the Policy Package

The examples presented so far are functional, but are lacking many features that would be useful in a real-world implementation. The following examples will present additional improvements to the policy package that will make it more practical and efficient to use.

A. Expanding the **policy** interface class

The following **policy** interface class adds additional methods for managing a policy.

```
interface class policy;

    pure virtual function string name();
    pure virtual function string type_name();
    pure virtual function string description();
    pure virtual function bit item_is_compatible(uvm_object item);
    pure virtual function void set_item(uvm_object item);
    pure virtual function policy copy();

endclass: policy

`endif
```

Fig. 15: Expanded **policy** interface class

The **name** , **description** , and **copy** methods are implemented by the policy (or policy macro) directly and provide reporting information useful when printing messages about the policy to the log for the former two, or specific behavior for making a copy for the latter. The remaining three methods are implemented by **policy_imp** and are shared by all policies.

B. Better type safety checking and reporting in **policy_imp** methods

Some of the benefits of above methods can be seen by examining the new **set_item** method used by **policy_imp** .

```
virtual function void set_item(uvm_object item);
    if (item == null) begin
        `uvm_error("policy::set_item()", "NULL item passed")

    end else if ((this.item_is_compatible(item)) && $cast(this.m_item, item)) begin
        `uvm_info(
            "policy::set_item()",
            $sformatf(
                "policy <%s> applied to item <%s>: %s",
                this.name(), item.get_name(), this.description()
```

```
            ),
            UVM_FULL
        )
        this.m_item.rand_mode( 1 );

    end else begin
        `uvm_warning(
            "policy::set_item()",
            $sformatf(
                "Item <%s> type <%s> is not compatible with policy <%s> type <%s>",
               item.get_name(), item.get_type_name(), this.name(), this.type_name()
            )
        )
        this.m_item = null;
        this.m_item.rand_mode( 0 );
    end
endfunction: set_item
```

Fig. 16: **set_item** method from **policy_imp**

The **set_item** method makes use of all the reporting methods to provide detailed log messages when **set_item** succeeds or fails. Additionally, the **item_is_compatible** is used before the **\$cast** method is called and the **rand_mode** state is kept consistent with the result of the cast.

C. Replacing **policy_list** *with* **policy_queue**

Eagle-eyed readers might have noticed the lack of presence of a **policy_list** class in any of the examples above after migrating to the improved policy interface. Rather, a single typedef is all that is necessary to manage policies in a class.

```
typedef policy policy_queue[$];
```

Fig. 17: The **policy_queue** typedef

The **policy_queue** type is capable of storing any policy that implements the **policy** interface. The default queue methods are sufficient for aggregating policies, and in practice we found that using policy queues as containers was more efficient than **policy_list** instances. For example, for functions expecting a **policy_queue** argument we can directly pass in array literals populated by calls to static constructor functions, allowing us to define, initialize, aggregate, and pass policies all in a single line of code!

D. Provide a common implementation for policy-enabled classes using the **policy_container** *interface and* **policy_object** *mixin*

Our solution encourages loosely coupled code by defining APIs that pass **policy_queue** arguments and hide a target object's policies queue from other classes.

```
interface class policy_container;

    // Queries
    pure virtual function bit has_policies();

    // Assignments
    pure virtual function void set_policies(policy_queue policies);
    pure virtual function void add_policies(policy_queue policies);
    pure virtual function void clear_policies();

    // Access
```

```
    pure virtual function policy_queue get_policies();

    // Copy
    pure virtual function policy_queue copy_policies();

endclass: policy_container

`endif
```

Fig. 18: The **policy_container** interface class

This interface class should be implemented by any class that needs to use policies. The method implementations can then be used to manipulate a policy queue as needed, including adding to, replacing, or clearing the queue, and returning a handle to the queue or a copy of the queue to use elsewhere.

The **policy_container** interface class allows for a base class implementation that can be used to manage constraints across the entire class hierarchy.

```
class policy_object #(type BASE=uvm_object) extends BASE implements
↪   policy_container;

    protected policy_queue m_policies;

    // Queries
    virtual function bit has_policies();
        // returns true/false based on size of m_policies
    endfunction: has_policies

    // Assignments
    virtual function void set_policies(policy_queue policies);
        // sets m_policies to a new queue of policies
    endfunction: set_policies

    virtual function void add_policies(policy_queue policies);
        // adds new policies to m_policies
    endfunction: add_policies

    virtual function void clear_policies();
        // clears m_policies
    endfunction: clear_policies

    // Access
    virtual function policy_queue get_policies();
        // return a handle to m_policies
    endfunction: get_policies

    // Copy
    virtual function policy_queue copy_policies();
        // a copy of m_policies
    endfunction: copy_policies
endclass: policy_object
```

Fig. 19: A **policy_object** base class implementation

This base class implements all the **policy_container** functionality (a complete example is available in Appendix **??**). Packaging a common implementation and interface into a mixin enables sharing a common implementation and API for policy

support among any classes that might benefit from the use of policies, as seen in the following example.

```
// Use policy_object for transactions
class base_txn extends policy_object #(uvm_sequence_item);

// Use policy_object for sequences
class base_seq #(type REQ=uvm_sequence_item, RSP=REQ) extends policy_object #(
↪  uvm_sequence#(REQ, RSP) );

// Use policy_object for configuration objects
class cfg_object extends policy_object #(uvm_object);
```

Fig. 20: Example classes using the **policy_object** base class

E. Protecting the policy queue enforces loosely coupled code

A subtle but significant additional benefit to using a base **policy_object** class along with the **policy_container** API is the ability to mark the container's policy queue as protected and prevent direct access to it.

The original implementation called **set_item** on each policy during the **pre_randomize** stage. That was necessary because the **policy_queue** was public, so there was nothing to prevent callers from adding policies without linking them to the target class.

Using an interface class and making the implementation private means the callers may only set or add policies using the available interface class routines, and because those routines are solely responsible for applying policies, they can also check compatibility (and filter incompatible policies) and call **set_item** immediately. This can be seen in the example **policy_object** implementation above, in the usage of the protected function **try_add_policy** .

By calling **set_item** when the policy is added to the queue, all policies will be associated with the target item automatically, so there is no need to do it during **pre_randomize** . This eliminates an easily-overlooked requirement for classes extending **policy_object** to make sure they call **super.pre_randomize()** in their local **pre_randomize** function.

VII. Conclusion

The improvements to the policy package presented in this paper provide a more robust and efficient implementation of policy-based constraints for SystemVerilog. The policy package is now capable of managing constraints across an entire class hierarchy, and the policy definitions are tightly paired with the class they constrain. The use of macros reduces the expense of defining common policies, while still allowing great flexibility in any custom policies necessary.

The complete policy package is available in Appendix **??**, with additional macro examples available in Appendix **??**. A functional package is also available for download [**?**] which can be included directly in a project to start using policies immediately.

References

[1] J. Dickol, "SystemVerilog Constraint Layering via Reusable Randomization Policy Classes," DVCon, 2015.
[2] J. Dickol, "Complex Constraints: Unleashing the Power of the VCS Constraint Solver," SNUG Austin, September 29, 2016.
[3] K. Vasconcellos, J. McNeal, "Configuration Conundrum: Managing Test Configuration with a Bite-Sized Solution," DVCon, 2021.
[4] C. McClish, "Bi-Directional UVM Agents and Complex Stimulus Generation for UDN and UPF Pins," DVCon, 2021.
[5] J. Vance, J. Montesano, M. Litterick, J. Sprott, "Be a Sequence Pro to Avoid Bad Con Sequences," DVCon, 2019.
[6] D. Mills, C. Haldane, "policy_pkg Source Code," https://github.com/DillanCMills/policy_pkg.

APPENDIX A
POLICY PACKAGE

A. *policy_pkg.sv*

```
`ifndef__POLICY_PKG__
`define __POLICY_PKG__

`include "uvm_macros.svh"

package policy_pkg;

    import uvm_pkg::*;

    `include "policy.svh"

    typedef policy policy_queue[$];

    `include "policy_imp.svh"

    `include "policy_container.svh"
    `include "policy_object.svh"

endpackage: policy_pkg

`endif
```

B. *policy.svh*

```
`ifndef__POLICY__
`define __POLICY__

interface class policy;

    pure virtual function string name();
    pure virtual function string type_name();
    pure virtual function string description();
    pure virtual function bit item_is_compatible(uvm_object item);
    pure virtual function void set_item(uvm_object item);
    pure virtual function policy copy();

endclass: policy

`endif
```

C. *policy_imp.svh*

```
`ifndef__POLICY_IMP__
`define __POLICY_IMP__

virtual class policy_imp #(type ITEM=uvm_object) implements policy;

    protected rand ITEM m_item;
```

```
7
8       pure virtual function string name();
9       pure virtual function string description();
10      pure virtual function policy copy();
11
12      virtual function string type_name();
13          return( ITEM::type_name );
14      endfunction: type_name
15
16      virtual function bit item_is_compatible(uvm_object item);
17          ITEM local_item;
18
19          return( (item != null) && ($cast(local_item, item)) );
20      endfunction: item_is_compatible
21
22      virtual function void set_item(uvm_object item);
23          if (item == null) begin
24              `uvm_error("policy::set_item()", "NULL item passed")
25
26          end else if ((this.item_is_compatible(item)) && $cast(this.m_item, item))
    ↪   begin
27              `uvm_info(
28                  "policy::set_item()",
29                  $sformatf(
30                      "policy <%s> applied to item <%s>: %s",
31                      this.name(), item.get_name(), this.description()
32                  ),
33                  UVM_FULL
34              )
35              this.m_item.rand_mode( 1 );
36
37          end else begin
38              `uvm_warning(
39                  "policy::set_item()",
40                  $sformatf(
41                      "Cannot apply policy '%0s' of type '%0s' to target object '%0s'
    ↪   of incompatible type '%0s'",
42                      this.name(), ITEM::type_name(), item.get_name(),
    ↪   item.get_type_name()
43                  )
44              )
45              this.m_item = null;
46              this.m_item.rand_mode( 0 );
47          end
48      endfunction: set_item
49
50  endclass: policy_imp
51
52  `endif
```

D. policy_container.svh

```
1   `ifndef__POLICY_CONTAINER__
2   `define __POLICY_CONTAINER__
```

```

interface class policy_container;

    // Queries
    pure virtual function bit has_policies();

    // Assignments
    pure virtual function void set_policies(policy_queue policies);
    pure virtual function void add_policies(policy_queue policies);
    pure virtual function void clear_policies();

    // Access
    pure virtual function policy_queue get_policies();

    // Copy
    pure virtual function policy_queue copy_policies();

endclass: policy_container

`endif
```

E. policy_object.svh

```
`ifndef__POLICE_OBJECT__
`define __POLICE_OBJECT__

class policy_object #(type BASE=uvm_object) extends BASE implements
↪  policy_container;

    protected policy_queue m_policies;

    // Queries
    virtual function bit has_policies();
        return( this.m_policies.size() > 0 );
    endfunction: has_policies

    // Assignments
    virtual function void set_policies(policy_queue policies);
        if( this.has_policies()) begin
            `uvm_warning( "policy", "set_policies() replacing existing policies" )
        end
        this.m_policies = {};
        foreach( policies(i) ) try_add_policy( policies[i] );
    endfunction: set_policies

    virtual function void add_policies(policy_queue policies);
        foreach( policies(i) ) try_add_policy( policies[i] );
    endfunction: add_policies

    virtual function void clear_policies();
        if ( this.has_policies() ) begin
            `uvm_info( "policy", $sformatf("clearing [%0d] policies from %s",
↪  this.m_policies.size(), this.get_name()), UVM_FULL)
        end
```

```
        this.m_policies = {};
    endfunction: clear_policies

    // Access
    virtual function policy_queue get_policies();
        return( this.m_policies );
    endfunction: get_policies

    // Copy
    virtual function policy_queue copy_policies();
        copy_policies = {};
        foreach( this.m_policies[i] ) begin
            copy_policies.push_back( this.m_policies[i].copy() );
        end
    endfunction: copy_policies

    protected function void try_add_policy( policy new_policy );
        if (new_policy.item_is_compatible(this)) begin
            `uvm_info( "policy", $sformatf("adding policy %s to %s",
↪   new_policy.name, this.get_name()), UVM_FULL)
            new_policy.set_item( this );
            this.m_policies.push_back( new_policy );
        end else begin
            `uvm_warning( "policy", $sformatf("policy %s not compatible with target
↪   %s", new_policy.name, this.get_name()))
        end
    endfunction: try_add_policy
endclass: policy_object

`endif
```

APPENDIX B
POLICY MACROS

A. *policy_macros.svh*

```
`ifndef__POLICY_MACROS__
`define __POLICY_MACROS__

//================================================
// Embedded POLICIES class macros
//================================================

`define start_policies(cls)                                                 \
class POLICIES;                                                             \
`m_base_policy(cls)

`define start_extended_policies(cls, parent)                                \
class POLICIES extends parent::POLICIES;                                    \
`m_base_policy(cls)

`define end_policies                                                        \
endclass: POLICIES

`define m_base_policy(cls)                                                  \
    typedef policy_imp#cls) base_policy;


//================================================
// Policy template macros
//================================================

`include "constant_policy.svh"
`include "fixed_policy.svh"
`include "member_policy.svh"
`include "range_policy.svh"

`endif
```

B. *constant_policy.svh*

```
`ifndef__CONSTANT_POLICY__
`define __CONSTANT_POLICY__

// Full policy definition
`define constant_policy(POLICY, FIELD, TYPE, CONST)                         \
`m_const_policy_class(POLICY, FIELD, TYPE, CONST)                           \
`m_const_policy_constructor(POLICY)

// Policy class definition
`define m_const_policy_class(POLICY, FIELD, TYPE, CONST)                    \
    class POLICY``_policy extends base_policy                               \
                                                                            \
        constraint c_policy_constraint {                                    \
            (m_item != null) -> (m_item.FIELD == TYPE'(CONST));              \
```

```
        }                                                                   \
                                                                            \
        function new();                                                     \
        endfunction: new                                                    \
                                                                            \
        virtual function string name();                                     \
            return (`"POLICY`");                                            \
        endfunction: name                                                   \
                                                                            \
        virtual function string description();                              \
            return (`"(FIELD==CONST)`");                                    \
        endfunction: description                                            \
                                                                            \
        virtual function policy copy();                                     \
            copy = new();                                                   \
        endfunction: copy                                                   \
    endclass: POLICY``_policy

// Policy constructor definition
`define m_const_policy_constructor(POLICY)                                  \
    static function POLICY``_policy POLICY();                               \
        POLICY = new();                                                     \
    endfunction: POLICY

`endif
```

C. fixed_policy.svh

```
`ifndef__FIXED_POLICY__
`define __FIXED_POLICY__

// Full policy definition
`define fixed_policy(POLICY, FIELD, TYPE, RADIX="%0p")                      \
`m_fixed_policy_class(POLICY, FIELD, TYPE, RADIX)                           \
`m_fixed_policy_constructor(POLICY, TYPE, RADIX)

// Policy class definition
`define m_fixed_policy_class(POLICY, FIELD, TYPE, RADIX="%0p")              \
    class POLICY``_policy extends base_policy                               \
        local TYPE          l_val;                                          \
        local string        l_radix=RADIX;                                  \
                                                                            \
        constraint c_policy_constraint {                                    \
            (m_item != null) -> (m_item.FIELD == TYPE'(l_val));             \
        }                                                                   \
                                                                            \
        function new(TYPE value, string radix=RADIX);                       \
            this.set_value(value);                                          \
            this.set_radix(radix);                                          \
        endfunction: new                                                    \
                                                                            \
        virtual function string name();                                     \
            return (`"POLICY`");                                            \
        endfunction: name                                                   \
```

```
                                                                    \
        virtual function string description();                      \
            return ({                                               \
                `"(FIELD==",                                        \
                $sformatf(l_radix, l_val),                          \
                `")`"                                               \
            });                                                     \
        endfunction: description                                    \
                                                                    \
        virtual function policy copy();                             \
            copy = new(l_val, l_radix);                             \
        endfunction: copy                                           \
                                                                    \
        virtual function void set_value(TYPE value);               \
            this.l_val = value;                                     \
        endfunction: set_value                                      \
                                                                    \
        virtual function TYPE get_value();                          \
            return (this.l_val);                                    \
        endfunction: get_value                                      \
                                                                    \
        virtual function void set_radix(string radix);              \
            this.l_radix = radix;                                   \
        endfunction: set_radix                                      \
                                                                    \
        virtual function string get_radix();                        \
            return (this.l_radix);                                  \
        endfunction: get_radix                                      \
    endclass: POLICY``_policy

// Policy constructor definition
`define m_fixed_policy_constructor(POLICY, TYPE, RADIX="%0p")       \
    static function POLICY``_policy POLICY(TYPE value, string radix=RADIX);  \
        POLICY = new(value, radix);                                 \
    endfunction: POLICY

`endif
```

D. ranged_policy.svh

E. set_policy.svh